AIMMS Language Reference - Node and Arc Declaration

This file contains only one chapter of the book. For a free download of the complete book in pdf format, please visit www.aimms.com or order your hard-copy at www.lulu.com/aimms.

Paragon Decision Technology B.V.
Schipholweg 1
2034 LS Haarlem
The Netherlands
Tel.: +31 23 5511512
Fax: +31 23 5511517

Paragon Decision Technology Inc.
500 108th Avenue NE
Ste. # 1085
Bellevue, WA 98004
USA
Tel.: +1 425 458 4024
Fax: +1 425 458 4025

Paragon Decision Technology Pte. Ltd.
80 Raffles Place
UOB Plaza 1, Level 36-01
Singapore 048624
Tel.: +65 9640 4182

Email: info@aimms.com
WWW: www.aimms.com

This documentation was typeset by Paragon Decision Technology B.V. using LATEX and the LUCIDA font family.

Chapter 16

Node and Arc Declaration

This chapter

This chapter discusses the special identifier types and language constructs that AIMMS offers to allow you to formulate network optimization problems in terms of nodes and arcs. In addition, it is illustrated how you can formulate an optimization problem that consists of a network combined with ordinary variables and constraints.

16.1 Networks

Networks

There are several model-based applications which contain networks and flows. Typical examples are applications for the distribution of electricity, water, materials, etc. AIMMS offers two special constructs, `ARCS` and `NODES`, to formulate flows and flow balances as an alternative to the usual algebraic constructs. Specialized algorithms exist for pure network problems.

Mixed formulations

It is possible to intermingle network constructs with ordinary variables and constraints. As a result, the choice between `ARCS` and `VARIABLES` on the one hand, and `NODES` and `CONSTRAINTS` on the other, becomes a matter of convenience. For instance, in the formulation of a flow balance at a node in the network you can refer to flows along arcs as well as to variables that represent import from outside the network. Similarly, you can formulate an ordinary capacity constraint involving both network flows and ordinary variables.

Flow keywords

It is assumed here that you know the basics of network flow formulations. Following are three flow-related keywords which can be used to specify a network flow model:

- `NetInflow`—the total flow into a node minus the total flow out of that node,
- `NetOutflow`—the total flow out of a node minus the total flow into that node, and
- `FlowCost`—the cost function representing the total flow cost built up from individual cost components specified for each arc.

The first two are always used in the context of a node declaration, while the third may be used for the network model declaration.

16.2 NODE declaration and attributes

Node attributes

Each node in a network has a number of associated incoming and outgoing flows. Unless stated otherwise, these flows should be in balance. Based on the flows specified in the model, AIMMS will automatically generate a balancing constraint for every node. The possible attributes of a NODE declaration are given in Table 16.1.

Attribute	**Value-type**	**See also page**
INDEX DOMAIN	*index-domain*	42, 207, 215
UNIT	*unit-valued expression*	45, 210
TEXT	*string*	19, 45
COMMENT	*comment string*	19
DEFINITION	*expression*	216
PROPERTY	NoSave, Sos1, Sos2, Level, Bound, ShadowPrice, RightHandSideRange, ShadowPriceRange	45, 212, 217

Table 16.1: NODE attributes

Nodes are like constraints

Nodes are a special kind of constraint. Therefore, the remarks in Section 14.2 that apply to the attributes of constraints are also valid for nodes. The only difference between constraints and nodes is that in the definition attribute of a node you can use one of the keywords NetInflow and NetOutflow.

NetInflow and NetOutflow

The keywords NetInflow and NetOutflow denote the net input or net output flow for the node. The expressions represented by NetInflow and NetOutflow are computed by AIMMS on the basis of all arcs that depart from and arrive at the declared node. Since these keywords are opposites, you should choose the keyword that makes most sense for a particular node.

Example

The following two NODE declarations show natural applications of the keywords NetInflow and NetOutflow.

```
NODE:
     identifier   : CustomerDemandNode
     index domain : (j in Customers, p in Products)
     definition   :
        NetInflow >= ProductDemanded(j,p) ;

NODE:
     identifier   : DepotStockSupplyNode
     index domain : (i in Depots, p in Products)
```

```
definition   :
    NetOutflow <= StockAvailable(i,p) + ProductImport(i,p);
```

The declaration of CustomerDemandNode(c,p) only involves network flows, while the flow balance of DepotStockSupplyNode(d,p) also uses a variable Product-Import(d,p).

16.3 ARC declaration and attributes

Arc attributes

Arcs are used to represent the possible flows between nodes in a network. From these flows, balancing constraints can be generated by AIMMS for every node in the network. The possible attributes of an arc are given in Table 16.2.

Attribute	**Value-type**	**See also page**
INDEX DOMAIN	*index-domain*	42
RANGE	*range*	207
DEFAULT	*constant-expression*	44, 209
FROM	*node-reference*	
FROM MULTIPLIER	*expression*	
TO	*node-reference*	
TO MULTIPLIER	*expression*	
COST	*expression*	
UNIT	*unit-valued expression*	210
PRIORITY	*expression*	210
NONVAR STATUS	*expression*	211
RELAX STATUS	*expression*	212
PROPERTY	NoSave, *numeric-storage-property*, Inline, SemiContinuous, ReducedCost, ValueRange, CoefficientRange	33, 45, 212
TEXT	*string*	19, 45
COMMENT	*comment string*	19

Table 16.2: ARC attributes

Arcs are like variables

Arcs play the role of variables in a network problem, but have some extra attributes compared to ordinary variables, namely the FROM, TO, FROM MULTIPLIER, TO MULTIPLIER, and COST attributes. Arcs do not have a DEFINITION attribute because they are implicitly defined by the FROM and TO attributes.

The FROM and TO attributes

For each arc, the FROM attribute is used to specify the starting node, and the TO attribute to specify the end node. The value of both attributes must be a reference to a declared node.

The MULTIPLIER attributes

With the FROM MULTIPLIER and TO MULTIPLIER attributes you can specify whether the flow along an arc has a gain or loss factor. Their value must be an expression defined over some or all of the indices of the index domain of the arc. The result of the expression must be positive. If you do not specify a MULTIPLIER attribute, AIMMS assumes a default of one. Network problems with non unit-valued MULTIPLIERs are called *generalized networks*.

FROM and TO MULTIPLIER

The FROM MULTIPLIER is the conversion factor of the flow at the source node, while the TO MULTIPLIER is the conversion factor at the destination node. Having both multipliers offers you the freedom to specify the network in its most natural way.

The COST attribute

You can use the COST attribute to specify the cost associated with the transport of one unit of flow across the arc. Its value is used in the computation of the special variable FlowCost, which is the accumulated cost over all arcs. In the computation of the FlowCost variable the component of an arc is computed as the product of the unit cost and the level value of the flow.

Graphically illustrated

In the presence of FROM and TO MULTIPLIERS, the drawing in Figure 16.1 illustrates

- the level value of the flow,
- its associated cost component in the predefined FlowCost variable, and
- the flows as they enter into the flow balances at the source and destination nodes (denoted by SBF and DBF, respectively).



Figure 16.1: Flow levels and cost from node i to node j

Semi-continuous arcs

You can only use the SemiContinuous property for arcs if you use an LP solver to find the solution. If you use the pure network solver integrated in AIMMS, AIMMS will issue an error message.

Example

Using the declaration of nodes from the previous section, an example of a valid arc declaration is given by

```
ARC:
     identifier   :  Transport
     index domain :  (i,j,p) | Distance(i,j)
     range        :  nonnegative
     from         :  DepotStockSupplyNode(i,p)
     to           :  CustomerDemandNode(j,p)
     cost         :  UnitTransportCost(i,j)  ;
```

Note that this arc declaration declares flows between nodes i and j for multiple products p.

16.4 Declaration of network-based mathematical programs

The FlowCost variable

If your model contains arcs and nodes, the special variable FlowCost can be used in the definition of the objective of your mathematical program. During the model generation phase, AIMMS will generate an expression for this variable based on the associated unit cost for each of the arcs in your mathematical program.

Pure network models

AIMMS will mark your mathematical program as a pure network, if the following conditions are met:

- your mathematical program consists of arcs and nodes only,
- all arcs are continuous and do not have one of the SOS or the SemiContinuous properties,
- the value of the OBJECTIVE attribute equals the variable FlowCost, and
- all MULTIPLIER attributes assume the default value of one,

For pure network models you can specify network as its TYPE.

Network versus LP solver

If your mathematical program is a pure network model, AIMMS will pass the model to a special network solver. If your mathematical program is a generalized network or a mixed network-LP problem, AIMMS will generate the constraints associated with the nodes in your network as linear constraints and use an LP solver to solve the problem. AIMMS will also use an LP solver if you have specified its type to be lp. You may assert that your mathematical program is a pure network model by specifying network as its type.

Example

A pure network model containing the arc and node declarations of the previous sections, but without the additional term ProductImport(d,p) in the node DepotStockSupplyNode(d,p), is defined by the following declaration.

```
MATHEMATICAL PROGRAM:
    identifier  : ProductFlowDecisionModel
    objective   : FlowCost
    direction   : minimize
    constraints : AllConstraints
    variables   : AllVariables
    type        : network ;
```

If the arc Transport(i,j) declared in the previous section is the only arc, then the variable FlowCost can be represented by the expression

```
    sum [(i,j,p), UnitTransportCost(i,j) * Transport(i,j,p)]
```

Note that the addition of the term ProductImport(i,p) in DepotStockSupply-Node(i,p) would result in a mixed network/linear program formulation, which requires an LP solver.